

Разработчик методических указаний для выполнения лабораторных работ доцент, к.ф.-м.н. Ласуков В.В.

Интерполяция с помощью многочленов

Задание

1. С помощью интерполяционных многочленов Лагранжа (или Ньютона) изучить распределение ошибки глобальной интерполяции в пределах таблицы заданной функции.
2. Выяснить влияние степени многочлена на ошибку интерполирования.
3. Используя "скользящий" интерполяционный многочлен, изучить влияние степени многочлена на ошибку интерполирования в зависимости от степени многочлена

Контрольные вопросы

1. Как распределяется ошибка интерполирования в пределах таблицы при глобальной интерполяции?
2. Как изменяется ошибка интерполирования внутри таблицы с ростом степени многочлена? Как она ведет себя на концах таблицы?
3. Как следует организовать построение интерполяционного многочлена при локальной интерполяции, чтобы минимизировать ошибку?

Создаем таблицу функции.

Здесь n - число интервалов разбиения таблицы

$$f(x) := \sin(x)$$

$$n \equiv 10$$

$$i := 0..n$$


$$a := 1$$

$$b := \pi$$

$$H := \frac{b - a}{n}$$

$$x_i := a + H \cdot i$$

$$y_i := f(x_i)$$



Ниже приведена программа для "скользящего" многочлена Лагранжа k-го порядка.


При $k = n$ имеет место глобальная интерполяция




$\text{Spl}(t, x, y, k) :=$	$N \leftarrow \text{last}(x)$ $i \leftarrow 0$ while $x_i < t \wedge i < N$ $i \leftarrow i + 1$ $\text{start} \leftarrow i + \text{trunc}\left(\frac{k}{2}\right) - k$ if $i + \text{trunc}\left(\frac{k}{2}\right) < N$ $\text{start} \leftarrow N - k$ otherwise $\text{start} \leftarrow 0$ if $\text{start} < 0$ for $i \in 0..k$ $Y_i \leftarrow y_{\text{start}+i}$ $X_i \leftarrow x_{\text{start}+i}$ $\sum_{i=0}^k Y_i \cdot \prod_{j=0}^k \text{if}\left(i = j, 1, \frac{t - X_j}{X_i - X_j}\right)$
-----------------------------	--

$$h := \frac{b - a}{500}$$

$t := a, a + h.. b$

Глобальная интерполяция





Замечание

Используя режим трассировки графиков, оцените величину максимальной погрешности интерполяции для многочленов различной степени.

Результаты оформите в виде таблицы:

степень многочлена	Максимальная ошибка	
	n = 5	n = 10
1		
2		
3		
4		
n		

Вычисление определенных интегралов

Цель работы

1. Вычислить приближенное значение определенного интеграла различными методами.
2. Получить зависимость интеграла от величины шага разбиения.
3. Сравнить оценки погрешности интегрирования для различных методов расчета.
4. Выбрать наиболее точный метод.

Порядок проведения работы и программа.

1. Задаем подынтегральную функцию, пределы интегрирования и получаем системное значение интеграла (В среде MathCad реализован метод Ромберга).

$$f(x) := \cos(x + x^3) \quad a := 0 \quad b := 1 \quad I0 := \int_a^b f(x) dx \quad I0 = 0.63344612$$

2. Составляем программу вычисления интеграла методом центральных прямоугольников

$$Ipr(m) := \begin{cases} h \leftarrow \frac{b-a}{2 \cdot m} \\ h \cdot \sum_{i=0}^{2 \cdot m - 1} f\left(a + i \cdot h + \frac{h}{2}\right) \end{cases} \quad Ipr(50) = 0.633461276$$

3. Составляем программу вычисления интеграла методом трапеций

$$Itr(m) := \begin{cases} h \leftarrow \frac{b-a}{2 \cdot m} \\ \frac{h}{2} \cdot \left[f(a) + f(b) + 2 \cdot \left(\sum_{i=1}^{2 \cdot m - 1} f(a + i \cdot h) \right) \right] \end{cases} \quad Itr(50) = 0.633415809$$

4. Составляем программу вычисления интеграла методом Симпсона

$$I_s(m) := \begin{cases} h \leftarrow \frac{b-a}{2 \cdot m} \\ \frac{h}{3} \left[f(a) + f(b) + 4 \cdot \sum_{i=1}^m f[a + h \cdot (2 \cdot i - 1)] + 2 \cdot \sum_{i=1}^{m-1} f(a + h \cdot 2 \cdot i) \right] \end{cases}$$


$$I_s(50) = 0.633446125$$

5. Исследуем зависимость значения интеграла от шага разбиения

$$k := 10 \dots 50$$

$$h_k := \frac{b-a}{2 \cdot k}$$

Зависимость от шага разбиения





6. Исследуем поведение апостериорной оценки погрешности (правило Рунге)

$$D_{pr_k} := \frac{|I_{pr}(2 \cdot k) - I_{pr}(k)|}{3}$$

$$D_{tr_k} := \frac{|I_{tr}(2 \cdot k) - I_{tr}(k)|}{3}$$

$$D_{s_k} := \frac{|I_s(2 \cdot k) - I_s(k)|}{15}$$





Обсуждение результатов

УКАЗАНИЕ.

Путем подбора параметров покажите, что в случае применения метода центральных прямоугольников и метода трапеций ошибки удовлетворительно аппроксимируются параболлами, а в методе Симпсона многочленами четвертой степени.

Почему методы прямоугольников и трапеций дают ошибки разного знака?

Численные методы решение нелинейных уравнений

Цель работы



1. Найти указанные корни заданного нелинейного уравнения тремя методами: методом хорд; методом касательных (метод Ньютона); методом простых итераций.
2. Для каждого метода исследовать влияние заданной точности ϵ на число потребовавшихся итераций.
3. Сравнить методы по скорости сходимости и выбрать наиболее быстро сходящийся.

Порядок выполнения работы и программа

1. Отделение корня.

Задаем функцию и строим ее обзорный график. Интервал отделения корня ищем в области определения функции постепенно сжимая область задания переменной. Отделяем заданный корень.

$$f(x) := e^{0.724 \cdot x} - 2.831 \cdot x \quad a := 2.5 \quad b := 3.5 \quad x := a, a + 0.2 .. b$$



2. Уточнение корня методом хорд

Задаем точность, положение неподвижной точки и число допустимых итераций.

УКАЗАНИЕ. Координаты точек на графике определяйте с помощью трассировки.

$$\varepsilon := 0.00000001$$

$$c := b$$

$$N := 50$$


Задаем последовательность приближенных значений корня.

```

U := | t0 ← a
      | t1 ← c - f(c) · (c - t0) / (f(c) - f(t0))
      | k ← 1
      | while |tk - tk-1| - ε ≥ 0 ∧ k < N + 2
      |   | tk+1 ← c - f(c) · (c - tk) / (f(c) - f(tk))
      |   | k ← k + 1
      | error(" k > N ") if k > N + 2
      | t

```

i := 0..last(U)



В качестве значения корня берем последнее значение последовательности U и заносим его в массив R. Число затраченных итераций заносим в массив T.

$$R_0 := U_{\text{last}(U)} \quad R = (2.915104541) \quad T_0 := \text{last}(U) \quad T = (17)$$

2. Уточнение корня методом Ньютона


За нулевое приближение корня принимаем неподвижную точку метода хорд.

Вычисляем производную функции

$$f_1(x) := \frac{d}{dx} f(x) \rightarrow .724 \cdot \exp(.724 \cdot x) - 2.831$$

Задаем последовательность приближенных значений корня.

```
V := | t0 ← c
      | t1 ← t0 - f(t0)/f1(t0)
      | k ← 1
      | while |tk - tk-1| - ε ≥ 0 ∧ k < N + 2
      |   | tk+1 ← tk - f(tk)/f1(tk)
      |   | k ← k + 1
      | t1 ← t0 - f(t0)/f1(t0)
      | t
      |
      | j := 0..last(V)
```



В качестве значения корня берем последнее значение последовательности V и заносим его в массив R . Число затраченных итераций заносим в массив T .

$$R_1 := V_{\text{last}(V)} \quad R = \begin{pmatrix} 2.915104541 \\ 2.915104543 \end{pmatrix} \quad T_1 := \text{last}(V) \quad T = \begin{pmatrix} 17 \\ 6 \end{pmatrix}$$

3. Уточнение корня методом простых итераций

Приводим уравнение к виду, удобному для итераций:

$$x = \phi(x)$$

Ищем

$\phi(x)$


в виде

$x + \lambda \cdot f(x)$


. Подбираем такое значение параметра λ , при котором выполняется достаточное условие сходимости метода. Результаты подбора контролируем с помощью графиков.

$$\lambda := -\frac{2}{5}$$

$$\phi(x) := x + f(x) \cdot \lambda$$



```
Z := | t0 ← b  
      | t1 ← φ(t0)  
      | k ← 1  
      | while |tk - tk-1| - ε ≥ 0 ∧ k < N + 2  
      |   | tk+1 ← φ(tk)  
      |   | k ← k + 1  
      | error("k>N") if k > N  
      | t  
k := 0..last(Z)
```



В качестве значения корня берем последнее значение последовательности Z и заносим его в массив R. Число затраченных итераций заносим в массив T.

$$R_2 := Z_{\text{last}(Z)} \qquad R = \begin{pmatrix} 2.915104541 \\ 2.915104543 \\ 2.915104544 \end{pmatrix}$$

$$T_2 := \text{last}(Z) \qquad T = \begin{pmatrix} 17 \\ 6 \\ 15 \end{pmatrix}$$

Обсуждение результатов и выводы

В отчете обязательно сформулируйте необходимые и достаточные условия сходимости методов. Обсудите замеченные трудности применения методов.

Приложение. Программа вычисления значения корня комбинированным методом хорд и касательных

$$\text{Kombi} := \begin{pmatrix} y_0 \\ x_0 \end{pmatrix} \leftarrow \text{if} \left[\left(\frac{f(b) + f(a)}{2} - f\left(\frac{a+b}{2}\right) \right) \cdot f(b) > 0, \begin{pmatrix} b \\ a \end{pmatrix}, \begin{pmatrix} a \\ b \end{pmatrix} \right]$$


$$k \leftarrow 0$$

$$\text{while } \left| \frac{x_k - y_k}{2} \right| > \varepsilon \wedge k < N$$

$$\left| \begin{array}{l} y_{k+1} \leftarrow y_k - \frac{f(y_k)}{f'(y_k)} \\ x_{k+1} \leftarrow y_{k+1} - f(y_{k+1}) \cdot \left(\frac{y_{k+1} - y_k}{f(y_{k+1}) - f(y_k)} \right) \\ k \leftarrow k + 1 \end{array} \right.$$

$$\begin{pmatrix} y \\ x \end{pmatrix}$$

$$m := 0.. \text{last}(\text{Kombig}) \qquad Y := \text{Kombig} \quad X := \text{Kombi}_1$$



Численные методы решения обыкновенных дифференциальных уравнений

Цель работы

1. Найти решение задачи Коши различными численными методами:

- методом Эйлера;
- методом прогноза и коррекции;
- методом Рунге-Кутты.

2. Сопоставить полученные решения.

Порядок проведения работы и программа.

Задаем правую часть исходного ДУ:

$$\frac{d}{dx} y(x) := 2 \cdot (x^2 + y(x))$$

$$f(x, y) := 2 \cdot (x^2 + y)$$

Точное решение

$$z(x) := 1.5 \cdot \exp(2 \cdot x) - x^2 - x - 0.5$$

Задаем начальные условия и пределы изменения переменной

$$n := 200$$

$$i := 1..n$$

$$a := 0$$

$$b := 6$$

$$h := \frac{b - a}{n}$$

$$x_0 := 0$$


$$x_i := x_0 + i \cdot h$$

$$y_0 := 1$$

Метод Эйлера

Заменяем исходное ДУ его дискретным аналогом


$$y_i := y_{i-1} + h \cdot f(x_{i-1}, y_{i-1})$$



Метод прогноза и коррекции

$$U_0 := y_0$$

$$U_i := U_{i-1} + \frac{h}{2} \left(f(x_{i-1}, U_{i-1}) + f(x_i, U_{i-1} + h \cdot f(x_{i-1}, U_{i-1})) \right)$$



Метод Рунге - Кутты 4-го порядка точности

$$k_0(x, y) := h \cdot f(x, y)$$

$$k_1(x, y) := h \cdot f\left(x + \frac{h}{2}, y + \frac{k_0(x, y)}{2}\right)$$

$$k_2(x, y) := h \cdot f\left(x + \frac{h}{2}, y + \frac{k_1(x, y)}{2}\right)$$

$$k_3(x, y) := h \cdot f(x + h, y + k_2(x, y))$$


$$v_0 := y_0$$

$$v_1 := v_{i-1} + \frac{1}{6} \cdot (k_0(x_{i-1}, v_{i-1}) + 2 \cdot k_1(x_{i-1}, v_{i-1}) + 2 \cdot k_2(x_{i-1}, v_{i-1}) + k_3(x_{i-1}, v_{i-1}))$$

$$d_{1i} := \frac{|z(x_i) - y_i|}{|1 + z(x_i)|}$$

$$d_{2i} := \frac{|z(x_i) - U_i|}{|1 + z(x_i)|}$$

$$d_{3i} := \frac{|z(x_i) - v_i|}{|1 + z(x_i)|}$$



Обсуждение результатов:

- Дать графическую иллюстрацию каждого метода.
- Выяснить, как влияет величина шага на погрешность каждого метода

Биномиальное распределение и его предельные формы

Методические указания и примерная программа проведения лабораторной работы (практического занятия) в среде MathCad по курсу “Теория вероятностей и математическая статистика”

1. Основные положения.

Пожалуй, самой распространенной вероятностной схемой, к которой сводится решение многих задач, является схема **повторения независимых испытаний в неизменных условиях**. Если в серии таких испытаний нас интересует только одно событие A , которое в каждом отдельном испытании может произойти с вероятностью p и не произойти с вероятностью $q = 1 - p$, то мы приходим к так называемой **схеме Бернулли**.

Любое случайное событие удобно описывать с помощью специальной случайной величины – **индикатора события (I)**. Индикатор события – это такая случайная величина, которая принимает значение, равное единице, если событие произошло и значение, равное нулю, если событие не произошло.

Введем случайную величину ξ , принимающую значения m , равное числу появления события A в серии из n испытаний. Множество элементарных событий для данного опыта (серии из n испытаний) состоит из всевозможных событий вида

$$A_i = \{ \xi = m_i; i = 1, 2, \dots, n; m = 0, 1, \dots, n \}$$

$$m_i = I_{i1} + I_{i2} + \dots + I_{in},$$

где I_i - индикатор i -го испытания.

Видно, что A_i можно рассматривать как последовательность, состоящую из m единиц и $n - m$ нулей. Вероятность такого события равна произведению вероятностей появления события ровно m раз и его не появления $n - m$ раз:

$$p^m \cdot q^{n-m}.$$

Число элементарных событий $\{ \xi = m \}$ равно числу сочетаний из n элементов по m . Напомним, что число сочетаний из m элементов по n рассчитывается по формуле:

$$C_m^n = \frac{n!}{m! \cdot (n-m)!}$$

При больших n и m вычисления факториалов вычисляются приближенно по формуле **Стирлинга** (1730 г.):

$$n! \approx n^n \cdot \exp(-n) \cdot \sqrt{(2 \cdot \pi \cdot n)}$$

(при $n = 10$ относительная погрешность вычислений не превышает 0.83%).

Окончательно, вероятность того, что в серии из n испытаний событие произойдет ровно m раз будет равно:

$$P\{ \xi = m \} = P(m, n) = C_m^n \cdot p^m \cdot q^{n-m}$$

Распределение случайной величины, задаваемое формулой (1) называется **биномиальным**. Свое название она получила, от того, что значения $P(m, n)$ являются членами в разложении $(p + q)^n$ по формуле **бинома Ньютона**:

$$(p + q)^n = \sum_{m=0}^n C_m^n \cdot p^m \cdot q^{n-m}$$

Поскольку $p + q = 1$, то

$$\sum_{m=0}^n C_m^n \cdot p^m \cdot q^{n-m} = 1$$

Эта формула просто иллюстрирует аксиому теории вероятностей, согласно которой вероятность достоверного события равна единице. Действительно, события

$$\{ \xi = m, m = 0, 1, \dots, n \}$$

составляют полную группу событий.

Исследуем характер зависимости $P(m, n)$ как функции от m . Для этого рассмотрим отношение:

$$Q = \frac{P(m+1, n)}{P(m, n)} = \frac{n-m}{m+1} \cdot \frac{p}{q}$$

Тогда интервал ее возрастания ($Q > 1$) определится неравенством:

$$m < n \cdot p - q,$$

а интервал убывания ($Q < 1$) неравенством:

$$m > n \cdot p + p.$$

Положение максимума функции m^* будет определяться соотношением:

$$np - q < m^* < np + p.$$

Это значение называется **наиболее вероятным**. Расстояние между границами в (2) равно ровно единице, поэтому, если $np - q$ дробное, то m^* будет единственным, если $(np - q)$ – целое, то наиболее вероятными будут два соседних значения: $(np - q)$ и $(np + p)$.

2. Числовые характеристики биномиального распределения.

2.1. Математическое ожидание и дисперсия.

По определению, математическое ожидание случайной величины вычисляется по формуле:

$$M\xi = \sum_{i=1}^n p_i \cdot x_i,$$

где

x_i - значения случайной величины ξ ,

p_i - вероятности событий $\{\xi = x_i\}$.

Для закона распределения случайной величины (1) мы получим:

$$M\xi = \sum_{m=0}^n m \cdot C_n^m \cdot p^m \cdot q^{n-m}.$$

Поскольку

$$m \cdot C_m^n = n \cdot C_{n-1}^{m-1},$$

то

$$\sum_{m=0}^n m \cdot C_n^m \cdot p^m \cdot q^{n-m} = n \cdot p \cdot \sum_{m=1}^{n-1} C_{n-1}^{m-1} \cdot p^{m-1} \cdot q^{n-m} = n \cdot p \cdot (p+q)^{n-1} = n \cdot p.$$

Окончательно:

$$M\xi = n \cdot p.$$

Для дисперсии, по определению, имеем:

$$D\xi = \sum_{i=1}^n (x_i - M\xi)^2 \cdot p_i.$$

С учетом (1) получим:

$$D\xi = \sum_{i=1}^n (x_i - M\xi)^2 \cdot p_i = \sum_{m=0}^n (m - n \cdot p)^2 \cdot C_n^m \cdot p^m \cdot q^{n-m} = n \cdot p \cdot q$$

2.2. Расчет числовых характеристик с помощью индикатора событий.

Аппарат индикаторов – чисто вероятностный метод – позволяет избежать громоздких формул суммирования.

Индикатор события I задается следующим законом распределения:

I_i	0	1	где I_i - значения индикатора; p_i - вероятность события в отдельном испытании.
p_i	1-p	p	

Математическое ожидание индикатора m_I и дисперсия D_I равны:

$$\left. \begin{aligned} m_I &= n \cdot p \\ D_I &= p \cdot q \end{aligned} \right\}$$

С помощью индикаторов отдельных испытаний I_i число появлений события A в схеме Бернулли выражается формулой:

$$m = \sum_{i=1}^n I_i$$

Вычислим математические ожидания и дисперсии от обеих частей выражения (3):

$$M[m] = M\left[\sum_{i=1}^n I_i\right] = \sum_{i=1}^n M[I_i] = n \cdot m_I = n \cdot p$$

$$D[m] = D\left[\sum_{i=1}^n I_i\right] = \sum_{i=1}^n D[I_i] = n \cdot D_I = n \cdot p \cdot q$$

3. Поведение при больших m и n .

3.1. Постановка задачи.

При больших m и n вычисления по формуле (1) становится затруднительным, так как, с одной стороны, факториалы больших чисел очень велики и возникает опасность переполнения разрядной сетки компьютера, с другой стороны, возведение в высокую степень чисел, меньше единицы (p и $q < 1$) грозит потерей числа (числа могут стать меньше машинного нуля). Использование формулы Стирлинга (1*) не снимает указанные выше вычислительные проблемы.

Очевидно, что возникающие при этом неопределенности вида $0 \times \infty$ следует устранять путем исследования предельного перехода в формуле (1) при

$$n, m \rightarrow \infty$$

Решение данной задачи связано с именами А.Муавра (1730 г.), П.Лапласа (1812 г.) и С. Пуассона (1837 г.).

В зависимости от предельного поведения вероятности события p , математического ожидания и дисперсии случайных величин, различают, в основном, два предельных распределения.

3.2. "Нормальное" приближение.

3.2.1. Локальная теорема Муавра - Лапласа.

Рассмотрим случай, когда n и np велики, т.е.

$$n \cdot p \xrightarrow{n \rightarrow \infty} \infty$$

Конечно, $P(m, n)$ по-прежнему будет иметь пик вблизи математического ожидания $m^* = n \cdot p$. Используя формулу Стирлинга (1*) для факториалов, перепишем выражение для $P(m, n)$ в следующем виде:

$$P(m, n) = \frac{1}{\sqrt{2 \cdot \pi \cdot n}} \cdot \left(\frac{m}{n}\right)^{-m-\frac{1}{2}} \cdot \left(\frac{n-m}{n}\right)^{-n-m-\frac{1}{2}} \cdot p^m \cdot (1-p)^{n-m} =$$

$$= \frac{1}{\sqrt{2 \cdot \pi \cdot n}} \cdot \exp \left[-\left(m + \frac{1}{2}\right) \cdot \ln\left(\frac{m}{n}\right) - \left(n - m + \frac{1}{2}\right) \cdot \ln\left(\frac{n-m}{n}\right) + \right.$$

$$\left. + m \cdot \ln(p) + (n-m) \cdot \ln(1-p) \right]$$

Введем новую переменную

$$x = m - n \cdot p,$$

которая равна отклонению m от своего математического ожидания. Полагая, что

$$\frac{x}{n \cdot p} \ll 1,$$

разлагаем (4) в ряд **Тейлора** по степеням x и оставляя только главные члены, получим:

$$P(m, n) \approx \frac{1}{\sqrt{(2 \cdot \pi \cdot n \cdot p \cdot q)}} \exp\left(-\frac{x^2}{2 \cdot n \cdot p \cdot q}\right)$$

Тем самым мы доказали **локальную предельную теорему Муавра**:

Если вероятность наступления некоторого события в n независимых испытаниях постоянна и равна p , то вероятность $P(m, n)$ того, что в этих испытаниях событие A наступит ровно m раз, удовлетворяет при $n \rightarrow \infty$ соотношению (5).

3.2.2. Интегральная теорема Муавра - Лапласа

При подсчете вероятности попадания случайной величины в заданный промежуток значений необходимо суммировать вероятности отдельных ее значений, принадлежащих заданному промежутку. В этом случае справедлива **интегральная теорема Лапласа**, которая утверждает, что

$$P\left(a \leq \frac{m - n \cdot p}{\sqrt{npq}} < b\right) \xrightarrow{n \rightarrow \infty} \frac{1}{\sqrt{2\pi}} \int_a^b \exp\left(-\frac{x^2}{2}\right) dx$$

3.3. Приближение Пуассона

Второй предел биномиального распределения, представляющий практический интерес, относится к случаю, когда при неограниченном увеличении числа испытаний математическое ожидание остается постоянным:

$$\left. \begin{array}{l} n \xrightarrow{n \rightarrow \infty} \theta \\ n \cdot p = \lambda = \text{Const} \end{array} \right\}$$

Если при $n \rightarrow \infty$ $p \rightarrow 1$, то перейдя к *противоположному событию*, мы получим тот же случай. Полагая $m \ll n$, получим при $n \rightarrow \infty$

$$\frac{n!}{(n-m)!} n(n-1)(n-2)\dots(n-m+1) \rightarrow n^m$$

$$(1-p)^{n-m} = (1-np/n)^{n-m} = \left(1 - \frac{\lambda}{n}\right)^{\frac{-\lambda \cdot n(n-m)}{\lambda \cdot n}} \rightarrow e^{-\lambda}$$

Следовательно,

$$P(m, n) = P(m, \lambda) = \frac{\lambda^m \exp(-\lambda)}{m!}$$

Полученное распределение вероятностей случайной величины называется **законом Пуассона**.

Распределение Пуассона имеет максимум вблизи

$$m_0 = [\lambda]$$

(знак $[x]$ обозначает целую часть числа x , меньшую или равную x).

Числовые характеристики распределения:

математическое ожидание

$$M\xi = \lambda$$

дисперсия

$$D\xi = np(1-p) \rightarrow \lambda$$

4. Практическое использование распределений.

Рассмотренные предельные распределения интересны не только тем, что их использование облегчает расчеты вероятностей событий в схеме Бернулли, но и

тем, что они описывают особые классы случайных величин, имеющих самостоятельное значение вне схемы Бернулли.

4.1. Нормальное распределение.

Нормальное распределение относится к числу наиболее распространенных и важных.

Оно часто используется для приближенного описания многих случайных явлений, когда интересующий нас результат складывается из большого количества независимых случайных факторов, среди которых нет сильно выделяющихся.

4.2. Распределение Пуассона.

Распределение Пуассона играет важную роль для описания "редких" событий в физике, теории связи, теории надежности, теории массового обслуживания и т.д. – там, где в течение определенного времени может происходить случайное число каких-то событий (радиоактивных распадов, телефонных вызовов, отказов оборудования, несчастных случаев и т.п.).

5. Задание к лабораторной работе (практическому занятию).

1. Задать функцию биномиального распределения и построить ее график.
2. Исследовать зависимость формы кривой и положения ее максимума от параметров n , p .
3. Задать функцию нормального (стандартного) распределения и построить ее график.
4. Представить на одном рисунке графики стандартного и биномиального распределений.
5. Оценить качество аппроксимации биномиального распределения нормальным.
6. Задать функцию распределения Пуассона и построить ее график.
7. Представить на одном рисунке графики Пуассоновского и биномиального распределений.
8. Оценить качество аппроксимации биномиального распределения распределением Пуассона.

6. Примерная программа проведения работы в среде MathCad

Биномиальное распределение и его предельные формы


Практическое занятие.

Цель занятия: исследовать характер предельного поведения биномиального распределения при различном предельном поведении математического ожидания и дисперсии.

Зададим функцию биномиального распределения и построим ее графики при различных значениях параметров.

$$P(m, n, p) := \frac{n! \cdot p^m \cdot (1-p)^{n-m}}{m! \cdot n - m!}$$

$$m := 0..n$$



В условиях справедливости локальной теоремы Муавра-Лапласа:

$$n \rightarrow \infty, p = \text{const}, n \cdot p, n \cdot p \cdot (1-p) \rightarrow \infty$$

биномиальное распределение переходит в нормальное с параметрами:

$$a := n \cdot p, s^2 := n \cdot p \cdot (1-p)$$

Зададим эти параметры как функции от n и p:

$$a(n, p) := n \cdot p$$

$$s(n, p) := \sqrt{n \cdot p \cdot (1-p)}$$

Зададим функцию нормального распределения

$$f(x, a, s) := \frac{\exp\left[-\frac{(x-a)^2}{2 \cdot s^2}\right]}{\sqrt{2 \cdot \pi \cdot s}}$$

Зададим вероятность = >

$$p1 := 0.8$$

Указание: изменяя


$$p1$$

проследит по графику за характером поведения распределений

$$m := 0..n$$

$$x := 0, 0.2..n$$

Нормальное и биномиальное распределения



Оценим качество нормального приближения с помощью функции качества $S(p)$:

$$S(p) := \sqrt{\left[\frac{1}{n} \sum_m (P(m, n, p) - f(m, a(n, p), s(n, p)))^2 \right]}$$

Оценим качество приближения Пуассона

Введем функцию распределения Пуассона :

$$Pu(x, \lambda) := \frac{\lambda^x}{x!} \cdot \exp(-\lambda)$$


Введем переменную

$$L(n, p) := n \cdot p$$


Указание: изменяя P проследит по графику за характером поведения распределения Пуассона

$$p := 0.3$$

Биномиальное и Пуассоновское




Построим графики функций $S(p)$ и оценим результаты аппроксимации
 $n = 100$
 $p := 0.01, 0.02 \dots 0.99$



Из этого графика видно, что нормальное распределение плохо работает как при малых, так и при больших значениях p . Введем "функцию качества" для распределения Пуассона

$$S(p) := \sqrt{\left[\frac{1}{n} \sum_m (P(m, n, p) - Pu(m, L(n, p)))^2 \right]}$$

$n \equiv 100$



Поведение $S(p)$ показывает, что Пуассоновское приближение лучше работает при малых значениях вероятности p

Указание: Изменяя n (в пределах 10 - 150) убедитесь, что ошибка аппроксимации уменьшается с ростом n .